

Application Note—Reverberation Algorithm(混响算法)

1. 概述

混响(reverb)算法是从网络上的各种信息资源中得到的，与Schroeder-Moorer模型相似，有三段：

- 一个早期的反射模拟器，将声音和七层以及延时的声音混合以模拟首先到达的声音。
- 一个室内模型模拟器，由一组四个平行的梳状滤波器组成，它们有不同的延时，增益以及内置的低通滤波器。
- 一个全通滤波器，通过关联梳状/低通滤波器来增加反射强度。

当使用了大部分可用的延时存储器单元时，要选择乘数和延时系数以提供一个好的反射值。在这种算法的设计，尤其是开发上我们的经验是非常丰富的。在听力测试中，例如，我们发现为了减少削峰，声音的全增益就会减小，混响的声音就会混浊不清。为了改善这个，就要增加反射的声音比率来清除这种声音，留下好听的混响声音，最后，再用一个信号发生器和振荡器来编程，结果就可以使增益减少来防止削峰。

2. 算法

现在开始，将左声道的25%的输入加到右声道的输入上，反过来也是。这是尽可能的考虑到在声音环境下实际发生的混频量。在那里，每个声道被独立的对待，即使在室内混响模型环境中也是如此。早期反射段由3321阶延时行组成，这些延时行的每一个分段的延时范围从955到3321个样点时长。（在48kHz时延时为20.7 到 69.2 ms，相应的尺寸22.4 到78 英尺，假定相对湿度为0%，温度70°F）。这些反射信号与原始信号相加一起送入四个梳状/低通滤波器中，这就是室内混响模型。这些滤波器的延时从2200到3744个样点时长，相应的尺寸为51.7 到 87.9英尺。基于上面的房间形状和收听者的位置，实际的房间尺寸应与滤波器相关联，且可以是上面那些长度的一半，因为混响的全循环模型是从一面墙到对面的墙然后再返回来。四个梳状-低通滤波器的输出要累加，结果要通过一个全通滤波器，依次加到“原声+早期反射”音频信号上，然后一起送到输出。

全通滤波器不符合普通的物理特性，但是在电子混响中却工作的很好。早期反射信号以及室内模型信号的总和在频率响应上将产生巨大的峰值和真实的下降，从而产生很好的回声效果。为了预防峰值中产生的削峰，增益必须降低，因此总的信噪比会受影响。全通滤波器可以帮助平滑某些peak，同时增加混响声音的密度。

；说明：混响算法程序举例

MEM earlyr 3321 ; 右声道早期反射
MEM combR1 2200
MEM lowpassR1 1
MEM combR2 2928
MEM lowpassR2 1
MEM combR3 2956
MEM lowpassR3 1
MEM combR4 3744
MEM lowpassR4 1

MEM allpassr 1201
MEM earlyl 3321 ; 左声道早期反射
MEM combL1 2200
MEM lowpass1L 1
MEM combL2 2928
MEM lowpass2L 1
MEM combL3 2956
MEM lowpass3L 1
MEM combL4 3744
MEM lowpass4L 1
MEM allpassl 1201
MEM temp1 1 ; 临时寄存器
MEM temp2 1 ; 临时寄存器

; 读取ADC，写入到早期反射内存中

RZP ADCR K=0.2
RAP ADCL K=0.05
WAP earlyr K=0.5
RAP earlyr+955 K=0.45
RAP earlyr+1055 K=0.06
RAP earlyr+1699 K=0.4
RAP earlyr+1867 K=0.3
RAP earlyr+1987 K=0.3
RAP earlyr+3055 K=0.13
RAP earlyr' K=0.12 ; 早期反射与原声相累加

; 梳状滤波器1
WAP temp1 K=0
RAP lowpassR1' k=.45
RAP combR1' k=.99
WAP lowpassR1 k=0
WZP temp2 k=.45
RAP temp1 k=.99
WAP combR1 k=0
; 梳状滤波器2
RZP temp1 K=.99
RAP lowpassR2' k=.49
RAP combR2' k=.99
WAP lowpassR2 k=0
WZP temp2 k=.42
RAP temp1 k=.99
WAP combR2 k=0
; 梳状滤波器3
RZP temp1 K=.99
RAP lowpassR3' k=.52
RAP combR3' k=.99
WAP lowpassR3 k=0

WZP temp2 k=.39
RAP temp1 k=.99
WAP combR3 k=0
; 梳状滤波器4
RZP temp1 K=.99
RAP lowpassR4' k=.54
RAP combR4' k=.99
WAP lowpassR4 k=0
WZP temp2 k=.38
RAP temp1 k=.99
WAP combR4 k=0

; 梳状滤波器的输出总和
RZP combR1' k=.2
RAP combR2' k=.2
RAP combR3' k=.2
RAP combR4' k=.2

; 全通
WAP temp2 k=0
RAP allpassr' k=.7
WZP allpassr k=0
RZP temp2 k=-.7
RAP allpassr' k=.51

; 加上早期反射
RAP temp1 k=0.999

; 写入输出端
WAP OUTR K=0

; 读取ADC，写入到早期反射内存中
RZP ADCL K=0.2
RAP ADCR K=.05
WAP earlyl K=.5
RAP earlyl+955 K=.45
RAP earlyl+955 K=.06
RAP earlyl+1699 K=.4
RAP earlyl+1867 K=.3
RAP earlyl+1987 K=.3
RAP earlyl+3055 K=.13
RAP earlyl' K=.12 ; 早期反射与原声相累加

; 梳状滤波器1
WAP temp1 K=0
RAP lowpass1L' k=.45
RAP combL1' k=.99


```
WAP lowpass1L k=0
WZP temp2 k=.45
RAP temp1 k=.99
WAP combL1 k=0
; 梳状滤波器2
RZP temp1 K=.99
RAP lowpass2L' k=.49
RAP combL2' k=.99
WAP lowpass2L k=0
WZP temp2 k=.42
RAP temp1 k=.99
WAP combL2 k=0
; 梳状滤波器3
RZP temp1 K=.99
RAP lowpass3L' k=.52
RAP combL3' k=.99
WAP lowpass3L k=0
WZP temp2 k=.39
RAP temp1 k=.99
WAP combL3 k=0
; 梳状滤波器4
RZP temp1 K=.99
RAP lowpass4L' k=.54
RAP combL4' k=.99
WAP lowpass4L k=0
WZP temp2 k=.38
RAP temp1 k=.99
WAP combL4 k=0

; 梳状滤波器的输出总和
RZP combL1' k=.2
RAP combL2' k=.2
RAP combL3' k=.2
RAP combL4' k=.2
;
; 全通
WAP temp2 k=0
RAP allpassl' k=.7
WZP allpassl k=0
RZP temp2 k=-.7
RAP allpassl' k=.51

; 加上早期反射
RAP temp1 k=0.999

; 写入输出端
WAP OUTL K=0
```

对右声道的处理框图





bd3201
Digital Reverb Engine

bd3201 Application Note—Reverberation Algorithm (中文版)

© 2006 BDNC ALL RIGHT RESERVED

www.bdnc.com

香港公司

比特联创(香港)有限公司
香港沙田科技大道西 6 号集成电路开发
中心 512-513 室

电话: 852-28542731 / 23916797
传真: 852-23916796
电子邮件: general@bdnc.com

北京公司

比特联创电子(北京)有限公司
北京市海淀区三里河路 21 号甘家口大厦
写字楼南座 1513 室, 邮编: 100037

电话: 86-10-88392985/88392986
传真: 86-10-88392980
电子邮件: bdncele@public.bta.net.cn

